Simulation Handouts

PortaCom is in the process to evaluate profitability and potential risk of loss for a new printer with innovative designs. Preliminary marketing and financial analysis show as follows:

Parameter (constant) Inputs			Probabilistic or variable inputs	
Selling Price Per Unit	$249		Direct Labor Cost/Unit	$45
Administrative Cost	$400,000		Parts Cost /Unit	$90
Advertising Cost	$600,000		First Year Demand (Units)	15,000

Base Case Scenario:

What if Analysis

	Worst	Best
Direct Labor Cost/Unit	$47	$43
Parts Cost /Unit	$100	$80
First Year Demand (Units)	1,500	28,500
Estimate First Year Profitability		

Random Number Generation and Probability Distribution

Direct Labor Cost/Unit	**Probability**	Parts Cost (Uniform Distribution)		Demand (Normal Distribution)	
$43	0.1	Smallest Value	$80	Mean ($\bar{X}$)	15,000
$44	0.2	Largest Value	$100	Std Deviation (s)	4,500
$45	0.4				
$46	0.2				
$47	0.1				
Sum	1.0				

	H	I	J	K	L	M	N	O
25	**Direct Labor Cost/Unit**			Random No. Intervals				
26	Cost $/Unit	Probability	Cumulative Probability	From	To	Cost ($)/Unit		
27	$43	0.1				$43		
28	$44	0.2				$44		
29	$45	0.4				$45		
30	$46	0.2				$46		
31	$47	0.1				$47		
32	**Sum**	**1**						
33								
34	**Parts Cost (Uniform Distribution)**							
35	RN	Part Cost						
36	0.6394							
37								
38	**Demand (Normal Distribution)**							
39	RN	Demand						
40	0.0836							
41								
42	Random Number or Probability RAND() is partically implemented in Excel@ formulas							
43			Direct labor		Parts		First Year	
44	Trial	RN	Cost/Unit	RN	Cost/Unit	RN	Demand	Net Profit $
45	1	0.3462		0.9050		0.0066		
46	2	0.9596		0.2272		0.0648		
47	3	0.1592		0.7728		0.9021		

Summary Statistics

25	A	B	C
26	**Summary Statistics**	Results	Excel@ Formula
27	Mean Profit		
28	Standard Deviation		
29	Min Profit		
30	Max Profit		
31	Number of Losses		
32	Number of Trials		
33	Probability of Loss		
34	Standard Error		
35	Margin of Error		
36	Lower 95% Limit		
37	Upper 95% Limit		

Solution for PortaCom Example

	A	B	C	D	E	F
18	**Direct Labor Cost/Unit**	**Probability**	**Parts Cost (Uniform Distribution)**		**Demand (Normal Distribution)**	
19	$43	0.1	Smallest Value	$80	Mean	15000
20	$44	0.2	Largest Value	$100	Std Deviation	4500
21	$45	0.4	Part Cost = SV+RAND()*(LV-SV)		Demand=NORMINV(RAND(),μ, σ)	
22	$46	0.2				
23	$47	0.1				
24	**Sum**	**1**				

(Sim.xlsx/PortCom)

	H	I	J	K	L	M	N	O	P
25	**Direct Labor Cost/Unit**			Random No. Intervals					
26	Cost $/Unit	Probability	Cumulative Probability	From	To	Cost ($)/Unit			
27	$43	0.1	0.10	0	0.10	$43			
28	$44	0.2	0.30	0.10	0.30	$44			
29	$45	0.4	0.70	0.30	0.70	$45			
30	$46	0.2	0.90	0.70	0.90	$46			
31	$47	0.1	1.00	0.90	1.00	$47			
32	**Sum**	**1**							
33									
34	**Parts Cost (Uniform Distribution)**								
35	RN	Part Cost	=Smallest + RAND()*(Largest -Smallest)						
36	0.6714	$93.43	=$D$19+H36*($D$20-$D$19)						
37	=RAND()								
38	**Demand (Normal Distribution)**								
39	RN	Demand	=NORMINV(RAND(),mean, std)						
40	0.3822	13,651	=NORMINV(H40,$F$19,$F$20)						
41	=RAND()								
42	Random Number or Probability RAND() is partically implemented in Excel@ formulas						=NORMINV(M45,$F$19,$F$20)		
43			Direct labor		Parts		First Year		
44	Trial	RN	Cost/Unit	RN	Cost/Unit	RN	Demand	Net Profit $	
45	1	0.3462	$45.00	0.9050	$98.10	0.0066	3,848	($592,547)	
46	2	0.9596	$47.00	0.2272	$84.54	0.0648	8,179	($39,276)	
47	3	0.1592	$44.00	0.7728	$95.46	0.9021	20,821	$1,280,843	
48			=VLOOKUP(I45,$K$27:$M$31,3)			=($B$4-J45-L45)*N45-$B$5-$B$6			
49					=$D$19+K45*($D$20-$D$19)				

50	Random Number or Probability RAND() is fully implemented in Excel@ formulas				
51		Direct labor	Parts	First Year	
52	Trial	Cost/Unit	Cost/Unit	Demand	Net Profit $
53	1	$46.00	$86.86	9,898	$149,509
54	2	$46.00	$97.45	18,947	$999,867
55	3	$45.00	$98.38	19,715	$1,082,295
56		=VLOOKUP(RAND(),$K$27:$M$31,3)			
57			=$D$19+RAND()*($D$20-$D$19)		
58				=NORMINV(RAND(),$F$19,$F$20)	
59					=($B$4-I53-J53)*K53-$B$5-$B$6

Solutions to Summary Statistics

	A	B	C
26	**Summary Statistics**	Result	Excel@ Formula
27	Mean Profit	$216,340	=AVERAGE(O55:O57)
28	Standard Deviation	$962,498	=STDEV(O55:O57)
29	Min Profit	($592,547)	=MIN(O55:O57)
30	Max Profit	$1,280,843	=MAX(O55:O57)
31	Number of Losses	2	=COUNTIF(O55:O57,"<0")
32	Number of Trials	3	=COUNT(O55:O57)
33	Probability of Loss	66.67%	=B31/B32
34	Standard Error	$555,698	=STDEV(O55:O57)/SQRT(B32)
35	Margin of Error	$1,089,149	=NORMINV(0.975,0,1)*B34
36	Lower 95% Limit	($872,809)	=B27-B35
37	Upper 95% Limit	$1,305,489	=B27+B35

The explanations of the following screen shots are given in the pages after these screen shots.



Screen Shots for Data/Data Analysis/Histogram and Data/Data Analysis/Descriptive Statistics in Excel@

The design of ProCom case simulation in Excel@

	A	B	C	D	E	F	G	H
1	**PortaCom Risk Analysis**						**Summary Statistics**	
2							Mean Profit	$711,165
3	Selling Price Per Unit		$249				Standard Deviation	515752
4	Administrative Cost		$400,000				Min Profit	($803,022)
5	Advertising Cost		$600,000				Max Profit	$2,325,981
6							Number of Losses	41
7	**Direct Labor Cost**			**Parts Cost (Uniform Distribution)**			Number of Trials	500
8	Lower	Upper		Smallest Value	$80		Probability of Loss	8.20%
9	Random No.	Random No.	Cost per Unit	Largest Value	$100		Standard Error	$23,065
10	0.0	0.1	$43				Margin of Error	$45,317
11	0.1	0.3	$44				Lower 95% Limit	$665,849
12	0.3	0.7	$45	**Demand (Normal Distribution)**			Upper 95% Limit	$756,482
13	0.7	0.9	$46	Mean	15000			
14	0.9	1.0	$47	Std Deviation	4500		Range	$3,129,003
15							Number of Groups	12
16				=NORMINV(RAND(),$E$13,$E$14)			Group Width	$300,000.00
17	**Simulation Trials**		=$E$8+RAND()*($E$9-$E$8)					
18		=VLOOKUP(RAND(),$A$10:$C$14,3)			=($C$3-B21-C21)*D21-$C$4-$C$5			
19		Direct labor	Parts	First Year			Profit Bins	($803,022)
20	Trial	Cost/Unit	Cost/Unit	Demand	Profit		Min	($600,000)
21	1	$45.00	$83.97	15,797	$896,103			($300,000)
22	2	$45.00	$80.88	11,549	$421,992			$0
23	3	$45.00	$95.68	19,582	$1,121,129			$300,000
24	4	$46.00	$90.27	18,227	$1,054,693			$600,000
25	5	$45.00	$80.47	14,477	$788,292			$900,000
26	6	$43.00	$90.63	13,578	$566,403			$1,200,000
27	7	$45.00	$85.85	22,854	$1,700,229			$1,500,000
28	8	$44.00	$93.81	18,128	$1,015,675			$1,800,000
29	9	$47.00	$94.05	15,440	$666,806			$2,100,000
30	10	$44.00	$91.00	13,933	$588,311			$2,400,000
31	11	$46.00	$92.97	20,223	$1,225,230			$2,700,000

	A	B	C	D	E	F	G	H
615	595	$43.00	$86.38	10,823	$294,634			
616	596	$44.00	$90.47	11,977	$371,721			
617	597	$46.00	$91.63	11,398	$269,397			
618	598	$44.00	$88.26	13,865	$618,550			
619	599	$46.00	$80.01	7,400	($89,923)			
620	600	$45.00	$88.45	16,000	$848,857			

Sim.xlsx/PortCom

The output of Descriptive Statistics and Histogram:

	A	B	C	D	E	F
1	*Net Profit*			*Net Profit*	*Frequency*	*Cumulative %*
2				-600000	8	1.60%
3	Mean	721,415	=AVERAGE(E121:E620)	-300000	9	3.40%
4	Standard Error	24,029	=STDEV(E121:E620)/SQRT(B15)	0	29	9.20%
5	Median	741,847		300000	55	20.20%
6	Mode	#N/A		600000	89	38.00%
7	Standard Deviation	537,314	=STDEV(E121:E620)	900000	119	61.80%
8	Sample Variance	2.8871E+11		1200000	104	82.60%
9	Kurtosis	0.39237		1500000	55	93.60%
10	Skewness	(0.22063)		1800000	17	97.00%
11	Range	3,097,796		2100000	13	99.60%
12	Minimum	(965,849)	=MIN(E121:E6200)	2400000	2	100.00%
13	Maximum	2,131,947	=MAX(E121:E620)	2700000	0	100.00%
14	Sum	3.6071E+08		More	0	100.00%
15	Count	500	=COUNT(A121:A620)	Number of Losses	46	=COUNTIF(E121:E620,"<0")
16	Largest(100)	1,134,089		Probability of Loss	9.20%	=E15/B15
17	Smallest(100)	299,340		Std Error of % Loss	1.293%	=SQRT(E16*(1-E16)/B15)
18	Confidence Level(95.0%)	47,211	=TINV(0.05,B15-1)*B4	Margin of Error (%)	2.53%	=NORMSINV(0.975)*E17
19	Lower 95% Limit	674,204	=B3-B18	Lower 95% Limit	6.67%	=E16-E18
20	Upper 95% Limit	768,627	=B3+B18	Upper 95% Limit	11.73%	=E16+E18
21						
22						
23	P					
24	M					



Now let us look at the whole process for the actual simulation of PorCom case:

Procedures for Simulation (you may want to set Excel@ to manual calculations with MS Office Button / Excel Options / Formulas / Workbook Calculation / Manual and Press F9 whenever you need update calculations or search Excel@ Help with auto calculation to get detailed instructions) :

1. Identify objectives and performance measures:
   a. Profitability as measured by Net Profit = (SP – DL Cost – Part Cost) * Demand – Overhead Cost
   b. Potential loss as measured by =Probability of Loss = No. of Losses/No. of Trials

2. Set values for parameter or constant inputs:
   a. Selling Price = $249
   b. Advertising cost = $400,000
   c. Administrative cost = $600,000

3. Set Nt = 600 from Cell E21 to Cell E620 , the number of replications equals to
   a. Ns = 500 from Cell E121 to Cell 620, the number of replications to collect performance measures
   b. Nw = 100 from Cell E21 to Cell E120, the number of replications of warm up

4. Calculate cumulative probabilities for Direct labor cost, and set up the table for From and To of random numbers, and Direct labor cost in three consecutive columns in the spreadsheet

5. Set up the Trial numbers in Cells A21 to A620 with 1in Cell A21 and 600 in Cell A620

6. Generate Month 1's values of random variables of
   a. Generate Direct labor cost per unit in Cell B21 with discrete probability distribution in Excel@ with =VLOOKUP(RAND(),$A$10:$C$14,3)
   b. Generate Part cost per unit in Cell C21 with uniform probability distribution in Excel@ with =$E$8+RAND()*($E$9-$E$8)
   c. Generate Yearly Demand in Cell D21 with normal probability distribution in Excel@ with =NORMINV(RAND(),$E$13,$E$14)

7. Calculate Net Profit in Cell E21 in Excel@ with =($C$3-B21-C21)*D21-$C$4-$C$5
8. Verify the correctness of each formula very carefully

9. Copy the formulas in the whole row 21 for Month 1 to rows 22 to row 620 or 599 rows of Nt

10. Calculate Summary statistics in Excel@
    a. Click Data/Data Analysis/Descriptive Statistics,
       i. With the Input Range of Cells from E121 to E620 for Net Profits,
       ii. Select New Worksheet Ply with name as PorCom2,
       iii. Click options for both Summary Statistics and Confidence Interval 95% for the mean and click OK to get the Descriptive Statistics Table in Worksheet PortCom2
       iv. You may format the values to make them more readable
       v. Results for Lower and Upper 95% Confidence Intervals are added. Please note the Confidence Interval (95%) in the output of Descriptive Statistics is the Margin of Error or t(α, n-1)* $s/\sqrt{n}$, where t(α, n-1) = TINV(0.05,n-1) or you may use t(α, n-1) $\approx$2 to get the approximate 95% confidence interval for the mean.

b. Use a range $3 million of Net Profits, use around 10 groups with a minimum from -$600,000 and a group width of $300,000, to get Frequency Bin Ranges for Net Profits as in Cells H19 to H31.

c. Click Data/Data Analysis/Histogram,
   i. With the Input Range of Cells from E121 to E620 for Net Profits,
   ii. With the Frequency Bin ranges in Cells H19 to H31,
   iii. Select New Worksheet Ply with name as PorCom3,
   iv. Click options Cumulative Percentage and Chart Output and click OK to see the results.

11. Validate the results with real business operations and carefully study the logic of the simulation

12. Analyze the results and provide recommendations

13. Output Analysis for PortCom Case: (Refer to the Histogram, Frequency Distribution and the Table of Descriptive Statistics on the next page)
   a. The estimate for the population mean Net Profit and its 95% confidence interval is given in Descriptive Statistics as follows:
      i. Mean ($\bar{X}$) = $721,415 and its 95% confidence intervals are given by $\bar{X} \pm t(0.05, 499) * s/\sqrt{n}$ from $674,204 to $768,627. Where The standard error = $s/\sqrt{n}$ = $24,029 and t(0.05, 499) = 1.9647. The 9%% approximate 95% confidence intervals can be given by using 2 to replace t(0.05, 499).
      ii. How to interpret the approximate 95% confidence interval for the mean net profit? We are 95% sure that the unknown true population mean net profit is between $674,204 and $768,627 or if we construct 100 95% confidence intervals as we did for this one, over 95 out of the 100 CIs would contain the unknown true population mean net profit.

   b. The estimated probability of loss and its 95% confidence interval can be derived as follows:
      i. The probability of loss is given by the Cumulative % = 9.20% when Net Profit is less than $0 or divide 46, the number of replications of Net Profit less than $0, by Ns = 500, the total number of replications in the simulation.
      ii. Use the equation $\bar{p} \pm z\left(1-\frac{\alpha}{2}\right) * \sqrt{\bar{p} * (1-\bar{p})/n}$ to compute the 95% confidence interval from 6.67% to 11.73% for the probability of loss, where z(0.975) = 1.96, $\bar{p}$ = 0.092 and n = 500.
      iii. How to interpret the 95% confidence interval for the probability of loss? We are 95% sure that the unknown true population proportion of loss is from 6.67% to 11.73% or if we construct 100 such 95% confidence intervals for the probability of loss as we did for this one, over 95 out of the 100 CIs would contain the unknown true population proportion of loss.

   c. What is the probability that the net profit will be more than $1.5 million?
      i. = 100% - the cumulaive % for Net Profit less than $1.5 million or 93.6% = 6.4%
      ii. The cumulative % for Net Profit less than $1.5 million is also shown in the Histogram as 93.6%.
      iii. A 95% confidence interval for the probabiltiy that the net profit will be more than $1.5 million can be constructed in the same way as to construct the 95% CI for the probabiltiy of loss.

(Anderson, Sweeney, Williams and Martin, 12e, pages 552 and 553

Butler Electrical Supply Company (BESC) sales home ventilation fan for $125 per unit with the unit cost of $75. The monthly demand for the fan is normally distributed with a mean of 100 units and a standard deviation of 20 units. BESC receives monthly delivery from its suppliers to replenish its inventory to a replenishment level of Q at the beginning of each month. A $15 inventory holding cost is charged for each unit that is not sold when the monthly demand is less than the replenishment level Q. A $30 stock-out cost is charge for each unit of shortage when the monthly demand is greater than the replenishment level Q. The objectives of the simulation are to assess the monthly net profit resulting from using a particular replenishment level Q and to assess the service level or the percentage of demand that will be satisfied.

Controllable input is the replenishment level Q
The probabilistic input is the monthly demand D
Output measures are the average monthly profit and the service level that equals to the ratio of total units sold to total demand.

Logics of Business Operations at Butler:

Case 1: $D \le Q$.

Gross Profit =

Holding Cost =

Net Profit =

Case 2: $D > Q$.

Gross Profit =

Shortage Cost =

Net Profit =

	A	B	C	D	E	F	G	H
1	Given	Inventory level Q	100		Demand Mean	100		
2		Unit Inventory holding cost	15		Demand Std	20		
3		Unit Shortage cost	30					
4		Sales price	125					
5		Unit cost	75					
6		Gross Profit /unit	50					
7								
8		1. If D <= Q, then D units are sold, Q - D units leftover						
9								
10		2. If D > Q, then Q units are sold, D - Q units are shotages						
11								
12		Excel formulas to compute Units sold, Units leftover and Units shortages						
13								
14		Demand (D)	Normal (100, 20)		=ROUND(NORMINV(RAND(),100,20),0)			
15		Units sold	If (D <= Q, D, Q)		=IF(B26<=$C$1,B26,$C$1)			
16		Units leftover	If (D <Q, Q - D, 0)		=IF(B26<$C$1,$C$1-B26,0)			
17		Units shortage	If (D > Q, D - Q, 0)		=IF(B26>$C$1,B26-$C$1,0)			
18								
19		Net Profit = Gross Profit * Units Sold - Inventory holding cost * Units leftover - Shortage cost * Units shortages						
20			=$C$6*C26-$C$2*D26-$C$3*F26					
21								
22		Inventory Cost	Unit Inventory holding cost * Units leftover			=$C$2*D26		
23		Shortage Cost	Unit shortage cost * Units shortage			=F26*$C$3		
24								
25	No.	Demand (D)	Units Sold	Units Leftover	Inventory Cost	Units Shortage	Shortage Cost	Net Profit
26	1	106	100	0	0	6	180	4820
27	2	83	83	17	255	0	0	3895
28	3	106	100	0	0	6	180	4820
29	4	115	100	0	0	15	450	4550
30	5	82	82	18	270	0	0	3830

Butler Class Exercise

1		**Butler inventory**						
2						Selling Price	$125	
3		Gross Profit per Unit		$50		Unit Cost	$75	
4		Holding Cost per Unit		$15				
5		Shortage Cost per Unit		$30		**Demand (Normal Distribution)**		
6						Mean	100	
7		**Replenishment Level Q**		***100***		Std Deviation	20	
8								
9		**Demand (Discrete Distribution)**			Random No. Intervals			
10		Demand	Probability	Cumulative Probability	From	To	Demand	
11	-2	60	0.02				60	
12	-1	80	0.15				80	
13	0	100	0.33				100	
14	1	120	0.33				120	
15	2	140	0.15				140	
16	3	159	0.02				159	
17		sum	1.00					
18								
19		**Demand (Discrete Distribution)**						
20		0.4286	100					
21								
22								
23		**Demand (Uniform Distribution)**			Lowest	60	Highest	160
24		0.4286	103					
25								
26								
27		**Demand (Normal Distribution)**						
28		0.9628	136					
29		0.2005	83					
30		0.6690	109					
31								
32		**Simulation**						
33		**Demand (Normal Distribution)**						
34		Total	317	295				
35		Month	Demand	sales	Gross Profit	Holding Cost	Shortage Cost	Net Profit
36		1	102	100	$5,000	0	61	$4,939
37		2	120	100	$5,000	0	610	$4,390
38		3	95	95	$4,732	80	0	$4,652

Butler Excel@ SImulation

	A	B	C	D	E	F	G	H
1	Butler inventory			Selling Price		$125		
2				Unit Cost		$75		
3	Gross Profit per Unit		$50					
4	Holding Cost per Unit		$15		**Summary Statistics**			
5	Shortage Cost per Unit		$30		Mean Profit	$4,271	=AVERAGE(G17:G316)	
6					Std Deviation	627	=STDEV(G17:G316)	
7	Replenishment Level Q		100		Min Profit	$1,606	=MIN(G17:G316)	
8					Max Profit	$4,997	=MAX(G17:G316)	
9	Demand (Normal Distribution)				Service Level	91.80%	=C15/B15	
10	Mean	100			Std Error	$36		
11	Std Deviation	20	=SUM(C17:C316)					
12		=SUM(B17:B316)			=IF(B17<=$C$7,$C$4*($C$7-B17),0)			
13			=IF(B17<=$C$7,B17,$C$7)					
14	Simulation	=NORMINV(RAND(),$B$10,$B$11)				=IF(B17>$C$7,$C$5*(B17-$C$7),0)		
15	Total	30267	27784	=$C$3*C17			=D17-E17-F17	
16	Month	Demand	sales	Gross Profit	Holding Cost	Shortage Cost	Net Profit	
17	1	128	100	$5,000	0	848	$4,152	
18	2	95	95	$4,739	78	0	$4,660	
19	3	80	80	$4,005	299	0	$3,706	
20	4	114	100	$5,000	0	413	$4,587	
21	5	63	63	$3,138	559	0	$2,579	

Sim.xlsx/Butler

Relationships among Q, Average Net Profit ($) and Service Level (%)

Replenishment Level Q	Average Net Profit ($)	Service Level (%)
100	$4,217	91.78%
105	$4,358	94.23%
110	4424	96.70%
115	4576	97.26%
120	4595	98.28%
125	4537	98.35%
130	4456	99.20%
135	4437	99.75%
140	4175	99.88%



	A	B	C	D	E	F	G
311	295	131	100	$5,000	0	939	$4,061
312	296	109	100	$5,000	0	272	$4,728
313	297	64	64	$3,209	537	0	$2,671
314	298	90	90	$4,485	154	0	$4,331
315	299	113	100	$5,000	0	400	$4,600
316	300	62	62	$3,086	574	0	$2,512

Dynamic Simulation Models: Discrete –Event Simulation
(Anderson, Sweeney, Williams and Martin, 12e, pages 557 and 560)

Wachovia Bank has one automated teller machine (ATM) in each of its branch. The bank wants to assess whether more than one ATM in each branch is necessary. The Bank established its service guidelines stating that the average waiting time for an ATM should be one minute or less.

Assume: customers arrivals follow an uniform distribution (0, 4) or = a + RAND()* (b – a) = RAND() * 4 and the ATM service times follow normal distribution (2, 0.5) with a mean of 2 minutes and a standard deviation of 0.5 minutes or =NORMINV(RAND(),μ,σ) = NORMINV(RAND(), 2, 0.5)

Logics of Wachovia Bank One ATM Operations:

Inter-arrival Time:
Arrival Time:
Service Start Time:
Waiting Time:
Service Time:
Completion Time:
Time in System:

Inter-arrival times: randomly generated with Uniform (0, 4) in Cell B21

Arrival time is the clock in time = last arrival (Clock in) time + current inter-arrival time
or in Cell C22: = C21 + B22

Service start time = the server available time:

If the current Arrival time is late than the last Clock out (Completion) time,
then the current Service start time = the current Arrival time,
else, the current Service start time = the last completion time

in Cell D22: = MAX(C22, G21) or =IF (C22 > G21, C22, G21)

Waiting time = Service start time – Arrival time

In Cell E22: = D22 – C22

Service time: randomly generated with Normal (2, 0.5)

Completion time is the clock out time = Service start time + Service time

In system time = Clock out (Completion) time – Clock in(Arrival) time

Wachovia Bank One ATM Class Exercise

	A	B	C	D	E	F	G	H
1	**Wachovia Bank One ATM Simulation Model**							RN
2								0.9814
3	**Interarrival Times (Uniform Distribution)**				**Service Times (Normal Distribution)**			0.2932
4	Smallest Value	0			Mean	2		0.5473
5	Largest Value	4			Std Deviation	0.5		0.079
6								0.306
7	**RN**	**Interarrival Time**						0.7634
8	0.3693							0.0714
9								0.1404
10								
11								
12	**RN**	**Service Time**						
13	0.7955							
14								
15								
16								
17	**Simulation**							
18								
19		Interarrival	Arrival	Service	Waiting	Service	Completion	Time
20	Customer	Time	Time	Start Time	Time	Time	Time	in System
21	1							
22	2							
23	3							
24	4							

Summary Statistics

Number Waiting:

Probability of Waiting:

Average Waiting Time:

Max Waiting Time:

Utilization of ATM:

Number Waiting > 1 min:

Solution to Class Exercise

	A	B	C	D	E	F	G	H
1	**Wachovia Bank One ATM Simulation Model**							RN
2								0.9814
3	**Interarrival Times (Uniform Distribution)**				**Service Times (Normal Distribution)**			0.2932
4	Smallest Value	0			Mean	2		0.5473
5	Largest Value	4			Std Deviation	0.5		0.079
6								0.306
7	**RN**	**Interarrival Time**						0.7634
8	0.3693	1.4772	=$B$4+RAND()*($B$5-$B$4)					0.0714
9								0.1404
10								
11								
12	**RN**	**Service Time**						
13	0.7955	2.4128	=NORMINV(A13,$F$4,$F$5)					
14								
15								
16								
17	**Simulation**		=C21+B22					
18		=$B$4+RAND()*($B$5-$B$4)			=D21-C21	=NORMINV(H3,$F$4,$F$5)		
19		Interarrival	Arrival	Service	Waiting	Service	Completion	Time
20	Customer	Time	Time	Start Time	Time	Time	Time	in System
21	1	3.926	3.926	3.926	0.000	1.728	5.654	1.728
22	2	2.189	6.115	6.115	0.000	1.294	7.409	1.294
23	3	1.224	7.339	7.409	0.070	2.359	9.768	2.429
24	4	0.286	7.624	9.768	2.143	1.461	11.228	3.604
25				=IF(C22>G21,C22,G21)			=D21+F21	
26	**Summary Statistics**							=G21-C21
27	Number Waiting			2	=COUNTIF($E$21:$E$24,">0")			
28	Probability of Waiting			50.00%	=D27/COUNT(E21:E24)			
29	Average Waiting Time			0.55	=AVERAGE(E21:E24)			
30	Max Waiting Time			2.14	=MAX(E21:E24)			
31	Utilization of ATM			60.93%	=SUM(F21:F24)/(G24)			
32	Number Waiting > 1 min			1	=COUNTIF(E21:E24,">1")			
33	Probability of Waiting > 1 min			0.2500	=D32/COUNT(E21:E24)			

Wachovia Bank One ATM Simulation

	A	B	C	D	E	F	G	H	I
1	Wachovia Bank One ATM Simulation Model								
2					**Summary Statistics**				
3	Interarrival Times (Uniform Distribution)				Number Waiting		595	=COUNTIF(E116:E1015,">0")	
4	Smallest Value	0			Probability of Waiting		66.11%	=I3/COUNT(E116:E1015)	
5	Largest Value	5			Average Waiting Time		1.88	=AVERAGE(E116:E1015)	
6					Max Waiting Time		11.09	=MAX(E116:E1015)	
7	Service Times (Normal Distribution)				Utilization of ATM		81.40%	=SUM(F116:F1015)/(G1015-G115)	
8	Mean	2			Number Waiting > 1 min		459	=COUNTIF(E116:E1015,">1")	
9	Std Deviation	0.5			Probability of Waiting > 1 min		0.5100	=I8/COUNT(E116:E1015)	
10				=IF(C17>G16,C17,G16)					
11			=C16+B17						
12	Simulation	=$B$4+RAND()*($B$5-$B$4)				=NORMINV(RAND(),$B$8,$B$9)			
13			=B16	=C16	=D16-C16		=D16+F16	=G16-C16	
14		Interarrival	Arrival	Service	Waiting	Service	Completion	Time	
15	Customer	Time	Time	Start Time	Time	Time	Time	in System	
16	1	2.16	2.16	2.16	0.00	1.21	3.36	1.21	
17	2	0.65	2.81	3.36	0.56	2.79	6.15	3.35	
18	3	2.30	5.11	6.15	1.05	2.24	8.39	3.28	
19	4	2.40	7.51	8.39	0.88	1.68	10.07	2.56	
20	5	2.48	9.99	10.07	0.08	1.90	11.97	1.98	
21	6	3.57	13.56	13.56	0.00	1.85	15.40	1.85	

1009	994	4.02	2558.05	2558.05	0.00	2.65	2560.70	2.65
1010	995	1.27	2559.32	2560.70	1.38	2.77	2563.47	4.14
1011	996	1.62	2560.94	2563.47	2.53	3.05	2566.52	5.58
1012	997	2.44	2563.38	2566.52	3.13	1.78	2568.30	4.91
1013	998	1.25	2564.63	2568.30	3.66	1.98	2570.28	5.65
1014	999	3.07	2567.70	2570.28	2.58	1.67	2571.95	4.25
1015	1000	2.57	2570.27	2571.95	1.68	2.26	2574.22	3.94

Sim.xlsx/ATM

Hungry Dawg Restaurants (Ragsdale, 5e Revised, page 563 with minor modifications in Excel@)

As an analyst at Hungry Dawg, Lisa Pon is asked to determine how much money the company needs to accrue in the coming year to pay for its employees' health insurance claims. The company is self insured, meaning that it pays health insurance claims with its own money although it contracts with an outside company to handle the administrative details of processing claims and writing checks. The money the company uses to pay claims comes from two sources: employee contributions or premiums deducted from employees' paychecks, and company funds. The company must pay whatever costs are not covered by employee contributions. Each employee covered by the health plan contributes $125 per month. However, the number of employees covered by the plan changes from month to month as employees are hired and fired, quit, or simply add or drop health insurance coverage. A total of 18, 533 employees were covered by the plan last month. The average monthly health claim per covered employee was $250 last month.

Hungry Dawg Restaurants Class Exercise (Rags12.xlsx/Figure12_2)

	A	B	C	D	E	F	G
1							
2			**Hungry Dawg Restaurants**				
3							
4	**Initial Conditions**				**Assumptions**		
5	**Number of Covered Employees**			**18,533**	**Increasing**	**2%**	**per month**
6	**Average Claim per Employee**			**$250**	**Increasing**	**1%**	**per month**
7	**Amount Contributed per Employee**			**$125**	**Constant**		
8							
9		**Number of**	**Employee**	**Avg Claim**	**Total**		**Company**
10	**Month**	**Employees**	**Contributions**	**per Emp.**	**Claims**		**Cost**
11	1						
12	2						

1. What is the equation to compute Number of Employees for Month 1?

Where:

X_o, X_1, …, and X_t = Number of Covered Employees in Month 0, 1, …, and t, respectively.

r_n = the monthly rate of changes of the number of covered employees.

In Excel@,

2. What is the equation to compute Monthly Employee Contributions (MEC)?

3. What is the equation to compute the Average Monthly Claim per Employee?

Where:
Y_o, Y_1, …, and Y_t = Average Monthly Claim per Employee in Month 0, 1, …, and t, respectively.
r_c = the monthly rate of changes of the average claim per employee.
In Excel@,

4. What is the equation to compute the Monthly Total Claims (MTC)?

5. What is the equation to compute the Monthly Company Cost (MCC)?

In Excel@,

6. What is the equation to compute the Total Company Cost (TCC)?

The results of the first 12 months are given in the Figure 12.2 as follows (Rags12.xlsx/Fig12_2)

9		Number of	Employee	Avg Claim	Total		Company
10	Month	Employees	Contributions	per Emp.	Claims		Cost
11	1	18,904	$2,363,000	$252.50	$4,773,260		$2,410,260
12	2	19,282	$2,410,250	$255.03	$4,917,488		$2,507,238
13	3	19,667	$2,458,375	$257.58	$5,065,826		$2,607,451
14	4	20,061	$2,507,625	$260.15	$5,218,869		$2,711,244
15	5	20,462	$2,557,750	$262.75	$5,376,391		$2,818,641
16	6	20,871	$2,608,875	$265.38	$5,538,746		$2,929,871
17	7	21,289	$2,661,125	$268.03	$5,706,091		$3,044,966
18	8	21,714	$2,714,250	$270.71	$5,878,197		$3,163,947
19	9	22,149	$2,768,625	$273.42	$6,055,980		$3,287,355
20	10	22,592	$2,824,000	$276.16	$6,239,007		$3,415,007
21	11	23,043	$2,880,375	$278.92	$6,427,154		$3,546,779
22	12	23,504	$2,938,000	$281.71	$6,621,312		$3,683,312
23		=ROUND($D$5*(1+$F$5)^A11,)			**Total Company Cost**		**$36,126,069**
24			=$D$7*B11	=ROUND($D$6*(1+$F$6)^A11,2)			=E11-C11
25					=D11*B11		=SUM(G11:G22)

Any change of any or all of these variables may lead to changes in Total Company Cost.

Hungry Dawg Restaurants Class Exercise

	A	B	C	D	E	F	G	H	I
1	**Hungry Dawg Restaurants Class Exercise**								
2									
3	**Number of Covered Employees (Uniform Distribution)**					**Average Claim per Employee (Normal Distribution)**			
4	**Initial Conditions**				**Problem Data**				
5	**Number of Covered Employees**			18,533	**Max Decrease**	3.0%	**Max Increase**	7%	**Uniform Distribution**
6	**Average Claim per Employee**			$250	**Mthly Increase**	1.0%	**Std Dev**	$3	**Normal Distribution**
7	**Amount Contributed per Employee**			$125	**Constant**				
8									
9	**RN**	**% Changes of No. of Covered Employees (Uniform Distribution)**							
10	0.6141	**3.14%**	=-$F$5+A10*($H$5-(-$F$5))						
11									RN
12									0.721
13									0.6229
14									0.4718
15	**RN**	**Average Claim per Employee (Normal Distribution)**							0.578
16	0.3163	**-1.4342**	=NORMINV(A16,0,$H$6)						0.8782
17									0.3937
18									0.0771
19									0.9868
20									0.5461
21									0.8575
22	For a claim in a month, 1) use RN to find out Changes by Std in Avg Claim,								
23	2) use D6 *(1+F6)^A11 to find the new monthly average claim								
24	3) adjust the new monthly average claim by plus or minus the number of Std given in 1)								
25				=$D$7*C28			=F28*C28		=G28-D28
26			**Number of**	**Employee**		**Avg Claim**	**Total**		**Company**
27	**Month**	**RN**	**Employees**	**Contributions**	**RN**	**per Emp.**	**Claims**		**Cost**
28	**1**	**0.8575**	19,566	$2,445,777	**0.0771**	248.23	$4,856,832		$2,411,056
29	**2**	**0.5461**	20,048	$2,505,967	**0.3937**	252.41	$5,060,164		$2,554,196
30	**3**	**0.9868**	21,425	$2,678,077	**0.8782**	263.55	$5,646,500		$2,968,423
31			=D5*(1+(-$F$5)+B28*($H$5-(-$F$5)))			=NORMINV(E28,D6*(1+$F$6)^A28,$H$6)			
32			=C28*(1+(-$F$5)+B29*($H$5-(-$F$5)))			=NORMINV(E29,F28*(1+$F$6)^A29,$H$6)			

(Rags12.xlsx/Fig12_9Blank)

Hungry Dawg Restaurants Final Version (Rags12.xlsx/Fig12_9WP)

	A	B	C	D	E	F	G	H	I	J
1										
2				Hungry Dawg Restaurants						
3										
4	**Initial Conditions**				**Problem Data**					
5	**Number of Covered Employees**			18,533	**Max Decrease**	3.0%	**Max Increase**	7%	**Uniform Distribution**	
6	**Average Claim per Employee**			$250	**Mthly Increase**	1.0%	**Std Dev**	$3	**Normal Distribution**	
7	**Amount Contributed per Employe**			$125	**Constant**					
8										
9		**Number of**	**Employee**	**Avg Claim**	**Total**		**Company**			
10	**Month**	**Employees**	**Contributions**	**per Emp.**	**Claims**		**Cost**			
11	**1**	19,320	$2,414,975	$252.24	$4,873,322		$2,458,347			
12	**2**	20,613	$2,576,592	$251.39	$5,181,777		$2,605,185			
13	**3**	21,982	$2,747,728	$259.39	$5,701,858		$2,954,130			
14	**4**	22,622	$2,827,760	$259.66	$5,873,938		$3,046,178			
15	**5**	23,214	$2,901,703	$267.16	$6,201,775		$3,300,072			
16	**6**	23,257	$2,907,064	$266.12	$6,189,090		$3,282,027			
17	**7**	23,172	$2,896,490	$267.69	$6,202,776		$3,306,286			
18	**8**	24,090	$3,011,278	$271.65	$6,544,212		$3,532,934			
19	**9**	23,886	$2,985,691	$274.87	$6,565,472		$3,579,781			
20	**10**	24,671	$3,083,856	$275.62	$6,799,748		$3,715,892			
21	**11**	24,508	$3,063,440	$276.12	$6,767,091		$3,703,651			
22	**12**	25,191	$3,148,930	$278.94	$7,026,980		$3,878,050			
23		=$D$5*(1-$F$5+RAND()*($F$5+$H$5))			**Total Company Cost**		$39,362,533			
24		Uniform Dist.		=NORMINV(RAND(),$D$6*(1+$F$6)^A11,$H$6)						
25			=$D$7*B11		=D11*B11		=E11-C11			

Topics to be covered:

1. What is simulation and where simulation is used?
2. Methods of Risk Analysis
   a. Best (most optimistic) case/worst (most pessimistic) case analysis
   b. What – if analysis
   c. Simulation
3. Random Number Generations
   a. General probability distribution with Probability Distribution Table. The Excel@ VLOOKUP(RAND(),From_To_Value_Arrays,3) is used to generate values of variables.
   b. Uniform distribution with a(lower limit) and b (upper limit). The following Excel@ formula is used to generate values of uniform distributed random variables:

      =a + RAND() *(b-a)



Demand (x) = lower limit (a) + RAND() (upper limit (b) – lower limit (a)) = 5 + RAND() (10 – 5)

   c. Poisson distribution with a mean arrival rate of λ. The following Excel@ formula is used to generate values of inter-arrival times of Poisson distributed random variables:

      = - 1/λ*LN(RAND())

   d. Exponential distribution with a mean service rate of μ. The following Excel@ formula is used to generate values of service times of Exponential distributed random variables:

      = - μ*LN(RAND())

   e. Bernoulli process with p as the probability of X = 1 and 1 – p as the probability of X = 0. The following Excel@ formula is used to generate values of Bernoulli process of random variables:

      =if(RAND()<p,1,0)

f. Normal distribution with a mean of μ and a standard deviation of σ. The following Excel@ formula is used to generate values of normal distributed random variables:

= NORMINV(RAND(), μ, σ) or = μ + NORMSINV(RAND())*σ



$$P(X < x) = NORMSDIST\left(z = \frac{x-\mu}{\sigma}\right) = NORMDIST(x, \mu, \sigma, TRUE)$$

$$x = \mu + z\big(P(X < x)\big) \times \sigma = \mu + NORMSINV(RAND()) \times \sigma = NORMINV(RAND(), \mu, \sigma)$$

Sales (x) = mean sales + NORMSINV(RAND()) × std sales = 10 + NORMSINV(RAND())× 2 =
= NORMINV(RAND(), meanSales, StdSales) = NORMINV(RAND(),10,2)

http://davidmlane.com/hyperstat/z_table.html

4. Simulation: Static (Monte Carlo) versus Dynamic (Discrete versus Continuous) Simulation

5. Mathematical model

$Y = f(X_1, X_2, \cdots, X_k)$ Where X_i is the variable i in the model for i = 1, 2, …, k

When the functional form of $f(\bullet)$ is not known and/or the values of the parameters of β_0, β_1, ⋯ , β_k are known or could not be found, simulation is one way to business decision making which evaluate the amount of reward and risk involved, the function of the uncertainty in the outcome of the decision and the magnitude of the potential gain or in the decision.

6. How to do simulation?
   a. Number of replications, over 1000

b. Data Analysis:
    i. Sample size (n)
    ii. Mean ($\bar{X}$)
    iii. Standard deviation (s)
    iv. Std Error ($s_{\bar{x}}$) = s/$\sqrt{n}$
    v. Min (Worst)
    vi. Max (Best)
    vii. (1-α)% CI for the true unknown population mean (μ): $\bar{y} \pm t(\alpha, n-1) * \mathrm{s}/\sqrt{n}$
    viii. (1-α)% CI for the true unknown population proportion (p):
    $\bar{p} \pm z\left(1-\frac{\alpha}{2}\right) * \sqrt{\bar{p} * (1-\bar{p})/n}$
    ix. Interpretation of the results
c. Common used simulation software: SLAM, Crystal Ball
d. Verification and validation